Markelov's task: http://www.artofproblemsolving.com/Forum/viewtopic.php?f=38&t=461077 (in english)
Carl Love: http://www.mapleprimes.com/posts/144499-Stunningly-Beautiful-Identity-Proved
Kitonum: http://www.mapleprimes.com/posts/144605-Another-Proof-Of-Stunningly-Beautiful-Identity

```
> restart; interface(version);
```

Classic Worksheet Interface, Maple 16.02, Windows, Nov 18 2012, Build ID 788210

```
> # prepare for readable presentation
  r:=1/2- 3*7^(1/3)+
    3/2*(-25+3*7^(2/3)+ 18*7^(1/3))^(1/3)+
  3/2*(-44+18*7^(1/3)+3*7^(2/3))^(1/3):
  data:=[
    a= cos(3/19*Pi)+cos(5/19*Pi)-cos(2/19*Pi),
    b= cos(1/19*Pi)+cos(7/19*Pi)-cos(8/19*Pi),
    c= (-cos(9/19*Pi)+cos(6/19*Pi)+cos(4/19*Pi))]:
```

```
> LHS:=
     surd(cos(3*pi/19) + cos(5*pi/19) + cos(17*pi/19), 3) +
     surd(cos(pi/19) + cos(7*pi/19) + cos(11*pi/19), 3) +
     surd(cos(9*pi/19) + cos(13*pi/19) + cos(15*pi/19), 3);
  #LHS:=convert(LHS, power) assuming 3 <= pi, pi <= Pi:
  LHS:=eval(LHS, pi=Pi):
  ``=convert(LHS, power);
  `` = a^(1/3)+b^(1/3)-c^(1/3);
  op(data);;
  ``;
  RHS:='r'^(1/3):

  'RHS= r^(1/3)', 'r'=r;
```

$$LHS := \mathrm{surd}\left(\cos\left(\frac{3\pi}{19}\right)+\cos\left(\frac{5\pi}{19}\right)+\cos\left(\frac{17\pi}{19}\right), 3\right)+\mathrm{surd}\left(\cos\left(\frac{\pi}{19}\right)+\cos\left(\frac{7\pi}{19}\right)+\cos\left(\frac{11\pi}{19}\right), 3\right)$$
$$+\,\mathrm{surd}\left(\cos\left(\frac{9\pi}{19}\right)+\cos\left(\frac{13\pi}{19}\right)+\cos\left(\frac{15\pi}{19}\right), 3\right)$$

$$= \left(\cos\left(\frac{3\pi}{19}\right)+\cos\left(\frac{5\pi}{19}\right)-\cos\left(\frac{2\pi}{19}\right)\right)^{(1/3)}+\left(\cos\left(\frac{\pi}{19}\right)+\cos\left(\frac{7\pi}{19}\right)-\cos\left(\frac{8\pi}{19}\right)\right)^{(1/3)}$$
$$-\left(-\cos\left(\frac{9\pi}{19}\right)+\cos\left(\frac{6\pi}{19}\right)+\cos\left(\frac{4\pi}{19}\right)\right)^{(1/3)}$$

$$= a^{(1/3)}+b^{(1/3)}-c^{(1/3)}$$

$$a=\cos\left(\frac{3\pi}{19}\right)+\cos\left(\frac{5\pi}{19}\right)-\cos\left(\frac{2\pi}{19}\right),\ b=\cos\left(\frac{\pi}{19}\right)+\cos\left(\frac{7\pi}{19}\right)-\cos\left(\frac{8\pi}{19}\right),$$
$$c=-\cos\left(\frac{9\pi}{19}\right)+\cos\left(\frac{6\pi}{19}\right)+\cos\left(\frac{4\pi}{19}\right)$$

$$RHS = r^{(1/3)},$$
$$r=\frac{1}{2}-3\,7^{(1/3)}+\frac{3\left(-25+3\,7^{(2/3)}+18\,7^{(1/3)}\right)^{(1/3)}}{2}+\frac{3\left(-44+18\,7^{(1/3)}+3\,7^{(2/3)}\right)^{(1/3)}}{2}$$

Task: prove that $LHS = RHS$, i.e. an identity between trigonometric radicals and purely akgebraic numbers.
Check first, that numerical that notations are correct:

```
> 'LHS = RHS'; evalf(%);
```

$$LHS = RHS$$

$$0.81573137171246 = 0.815731371712444$$

The LHS is abbreviated as an algebraic expression (cf Kitonum). One can use resolvents to see, that sums of (roots of) algebraic numbers are algebraic numbers again and that way one gets a polynomial, for which the LHS is a root (independend of feeding a,b,c by specific constants) by construction (even if Maple may fail to recognize it).

```
> fa,fb, fc := x -> a-x^3, x -> b-x^3, x -> -c-x^3;
  ``;
  'fab(x)'='resultant(fa(y), fb(x-y), y)';
  collect(rhs(%), x): sort(%,x):
  fab:=unapply(%, x):
  #'fab(a^(1/3)+b^(1/3))': '%' = expand(%);

  'g(x)' = 'resultant(fab(y), fc(x-y), y)';
  collect(rhs(%), x): sort(%,x):
  g:=unapply(%, x):
  ``;
  'g(a^(1/3)+b^(1/3)-c^(1/3))': '%' = expand(%);
  ``;
  deg:=degree(g(x),x):
  sum( coeff(g(x),x, j )*x^(j), j = deg-6 .. deg) + `lower degrees ...`:
  sort(%,x):
  'g(x)' = %;
```

$$fa, fb, fc := x \to a - x^3, x \to b - x^3, x \to -c - x^3$$

$$fab(x) = resultant(fa(y), fb(x - y), y)$$

$$g(x) = resultant(fab(y), fc(x - y), y)$$

$$g(a^{(1/3)} + b^{(1/3)} - c^{(1/3)}) = 0$$

$$g(x) = -x^{27} + (-9 c + 9 b + 9 a) x^{24} + (-9 a c - 36 c^2 + 9 a b - 36 a^2 - 36 b^2 - 9 b c) x^{21} + \text{lower degrees ...}$$

Staring at the (suppressed) result one 'sees': it is a polynomial in x^3 and of degree 27.

So rewrite it as polynomial in X = x^3 of degree 9. And feed that with the trigonometric values for a,b,c:

```
> G:= X -> 'eval'(g(X^(1/3)), data);
  eval(g(X^(1/3)), data): combine(%): collect(%, X): sort(%,X):
  G:=unapply(%, X):
  ``;
  'type(G(X), polynom)':
  'degree(G(X), X)': '%%' = %%, '%'=%, x^3=X ;
  #'type(G(X), polynom)': '%'=%;
  #'degree(G(X), X)': '%'=%, x^3=X ;
  ``;
  deg:=degree(G(X),X):
  sum( coeff(G(X),X, j )*x^(j), j = deg-1 .. deg) + `lower degrees ...`:
  sort(%,x):
  'G(X)' = %;
```

$$G := X \to \text{'eval'}(g(X^{(1/3)}), data)$$

$$type(G(X), polynom) = true, degree(G(X), X) = 9, x^3 = X$$

$$G(X) = -x^9 + \left(9\cos\left(\frac{9\pi}{19}\right) - 9\cos\left(\frac{6\pi}{19}\right) - 9\cos\left(\frac{4\pi}{19}\right) + 9\cos\left(\frac{\pi}{19}\right) + 9\cos\left(\frac{7\pi}{19}\right) - 9\cos\left(\frac{8\pi}{19}\right) + 9\cos\left(\frac{3\pi}{19}\right) + 9\cos\left(\frac{5\pi}{19}\right) - 9\cos\left(\frac{2\pi}{19}\right)\right) x^8 + \text{lower degrees ...}$$

```
> 0='G(X)';
  X=fsolve(%);
  x=rhs(%)^(1/3);
```

$$0 = G(X)$$
$$X = 0.542802070311500$$
$$x = 0.815731372189082$$

Then a real root can be computed numerically, which up to rounding errors leads to the desired value of the RHS. Now dare to solve it symbolically.

The answer is in terms of indexed RootOf (because we have higher degree). But astonishingly through that Maple now has found a polynomial over the integers for it - it silently has simplified the trigonometric coefficients (I failed trying that with the usual commands) !

```
> 0='G(X)';
  [solve(%,X)]: %[1], `... and higher indices`;

  [%][1]:
  op(1,%): subs(_Z=X, %): sort(%):
  H:=unapply(%, X);
```

$$0 = G(X)$$

$$\text{RootOf}(-3429500 + 19905597\,\_Z - 1152\,\_Z^8 + 256\,\_Z^9 + 1187856\,\_Z^5 + 33408\,\_Z^7 + 42581388\,\_Z^3 - 43229286\,\_Z^2 + 865632\,\_Z^6 - 17590320\,\_Z^4, \text{index} = 1), \text{... and higher indices}$$

$$H := X \to 256 X^9 - 1152 X^8 + 33408 X^7 + 865632 X^6 + 1187856 X^5 - 17590320 X^4 + 42581388 X^3 - 43229286 X^2 + 19905597 X - 3429500$$

Extract that polynomial H, which is of degree = 9 as well. If the coincidence is true, then G and H must be the same, if being normed to have leading coefficient = 1. Which is true. And astonishingly the command **is** can 'prove' it:

```
> 'H(X)/256 =- G(X)';
  is(%);
```

$$\frac{1}{256} H(X) = -G(X)$$

true

Now for the polynomial H one verifies that RHS^3 is a root and by positivity (left to the reader) it follows that RHS is a root of g, for which

```
> 'H(RHS^3)': '%'= evala(%);
```

$$H(RHS^3) = 0$$

Using Sturm chains there is only 1 root over the Reals, so $LHS^3 = RHS^3$, so they are equal. By positivity (left to the reader) it follows $LHS = RHS$.

```
> 'sturm(H(X),X,-infinity,infinity)': '%'=%;
```

$$\text{sturm}(H(X), X, -\infty, \infty) = 1$$